Statistical Methods in Epidemiology
Lab 3.
Interactions

I. Interactions- Apporach 1
In the genetic dataset, data of 328 records of individuals are stored. The outcome of interest is hypertension, (variable hyper). We want to examine whether the development of hypertension is associated with a person’s Body Mass Index, a score indicating the dedication to Mediterranean diet, categorised in two groups and, most importantly, whether and how these two characteristics interact with each other. 
1. Open the datafile genetic and inspect it. Check the variables’ coding. Is it meaningful in the context of an interaction exploration?
use "hyper", clear
desc
sum 

In order to study the interaction between BMI and dedication to Mediterranean diet, we need to generate a single variable that combines information from both the examined variables. The new variable should be equal to 1 for both factors being absent and 4 for both factors being present. 
gen     cat_bmi_nomd=1 if bmi==0 & nomd==0
replace cat_bmi_nomd=2 if bmi==1 & nomd==0
replace cat_bmi_nomd=3 if bmi==0 & nomd==1
replace cat_bmi_nomd=4 if bmi==1 & nomd==1


2. After creating the cat_bmi_nomd variable, we can study the interaction following the first method and we are going to estimate interactions in additive form.
logistic hyper i.cat_bmi_nomd

3. After fiting the logistic regression model, we can construct the “relative excess risk for interaction” (RERI) index. This can be done by combining the model estimate using the nlcom Stata command for the nonlinear combinations of estimators
nlcom:   exp(_b[4.cat_bmi_nomd])-exp(_b[2.cat_bmi_nomd])-exp(_b[3.cat_bmi_nomd])+1
we can give a name to the combined estimate:
nlcom  m1_RERI:   exp(_b[4.cat_bmi_nomd])-exp(_b[2.cat_bmi_nomd])-exp(_b[3.cat_bmi_nomd])+1

4. In a similar way the Synergy (S) index can be estimated. 

nlcom  m1_ln_SYN: ln(exp(_b[4.cat_bmi_nomd])-1)-ln(exp(_b[2.cat_bmi_nomd])+exp(_b[3.cat_bmi_nomd])-2) , post

The point estimate, as well as the lower and upper limits of a 95% confidence interval can be  assigned to scalars, so that we can use them later in our analysis. 

scalar m1_Syn_index       =exp(_b[m1_ln_SYN])
scalar m1_Syn_index_low95 =exp(_b[m1_ln_SYN]-invnormal(0.975)*_se[m1_ln_SYN])
scalar m1_Syn_index_high95=exp(_b[m1_ln_SYN]+invnormal(0.975)*_se[m1_ln_SYN])



mat define Model1_Synergy_index=(m1_Syn_index, m1_Syn_index_low95, m1_Syn_index_high95)
mat rown   Model1_Synergy_index= Syn_index
mat coln   Model1_Synergy_index= S_index  S_low95 S_high95
mat list   Model1_Synergy_index


5. Now we will get corresponding indices for multiplicative interactions:


logistic hyper i.cat_bmi_nomd
nlcom  m1_ln_mult_int: _b[4.cat_bmi_nomd]-_b[2.cat_bmi_nomd]-_b[3.cat_bmi_nomd] , post
scalar m1_mult_interaction        = exp(_b[m1_ln_mult_int])
scalar m1_mult_interaction_low95  = exp(_b[m1_ln_mult_int]-invnormal(0.975)*_se[m1_ln_mult_int])
scalar m1_mult_interaction_high95 = exp(_b[m1_ln_mult_int]+invnormal(0.975)*_se[m1_ln_mult_int])


mat define Model1_mult_interaction=(m1_mult_interaction, m1_mult_interaction_low95, m1_mult_interaction_high95)
mat rown   Model1_mult_interaction= mult_interaction 
mat coln   Model1_mult_interaction= mult_interaction  m_int_low95 m_int_high95
mat list   Model1_mult_interaction


II. Interactions- Apporach 2

1. In this exercise, we are going to explore interaction, too. This time we will follow the classic approach (i.e. we will fit a logistic model that includes both bmi, nomd and a term for their interaction)
gen bmi_nomd = bmi*nomd


logistic hyper  bmi nomd bmi_nomd

2. Using the corresponding formulas for the additive interaction, we get: 
nlcom m2_RERI:   exp(_b[bmi]+_b[nomd]+_b[bmi_nomd])-exp(_b[bmi])-exp(_b[nomd])+1
nlcom m2_ln_SYN: ln(exp(_b[bmi]+_b[nomd]+_b[bmi_nomd])-1)-ln(exp(_b[bmi])+exp(_b[nomd])-2) ,post
scalar m2_Syn_index       =exp(_b[m2_ln_SYN])
scalar m2_Syn_index_low95 =exp(_b[m2_ln_SYN]-invnormal(0.975)*_se[m2_ln_SYN])
scalar m2_Syn_index_high95=exp(_b[m2_ln_SYN]+invnormal(0.975)*_se[m2_ln_SYN])


mat define Model2_Synergy_index=(m2_Syn_index, m2_Syn_index_low95, m2_Syn_index_high95)
mat rown   Model2_Synergy_index= Syn_index
mat coln   Model2_Synergy_index= S_index  S_low95 S_high95
mat list   Model2_Synergy_index

III. Erroneous approach
1. In the previous, correct approach, we made sure that both factors under study took values on the same directions, i.e. baseline level indicated in both factors the lower risk category. Let us now see what would have happened if instead of the correct approach we followed so far, we had set as the baseline category of the four-level variable that combines bmi and diet for the second one (i.e. those who have low BMI but have low score for the dedication to the Mediterranean diet)  

char cat_bmi_nomd [omit] 2
and fit the same logistic regression model

xi: logistic hyper i.cat_bmi_nomd
Estimate RERI and Synergy indices, using same formulas as before:

nlcom  m1_err_RERI:   exp(_b[_Icat_bmi_n_4])-exp(_b[_Icat_bmi_n_3])-exp(_b[_Icat_bmi_n_1])+1
nlcom  m1_err_ln_mult_int: _b[_Icat_bmi_n_4]-_b[_Icat_bmi_n_3]-_b[_Icat_bmi_n_1] , post
scalar m1_err_mult_interaction        = exp(_b[m1_err_ln_mult_int])
scalar m1_err_mult_interaction_low95  = exp(_b[m1_err_ln_mult_int]-invnormal(0.975)*_se[m1_err_ln_mult_int])
scalar m1_err_mult_interaction_high95 = exp(_b[m1_err_ln_mult_int]+invnormal(0.975)*_se[m1_err_ln_mult_int])



mat define Model1_err_mult_interaction=(m1_err_mult_interaction, m1_err_mult_interaction_low95, m1_err_mult_interaction_high95)
mat rown   Model1_err_mult_interaction= mult_interaction 
mat coln   Model1_err_mult_interaction= mult_interaction  m_int_low95 m_int_high95
mat list   Model1_err_mult_interaction


xi: logistic hyper i.cat_bmi_nomd
lincom  _b[_Icat_bmi_n_4]-_b[_Icat_bmi_n_3]-_b[_Icat_bmi_n_1]


2. Please try a classic logistic regression, now considering a different coding for the mediterranean diet dedication considering a low scroe as the refrnece category (i.e nomd: dedication 1: ≥5, no dedication 0: <5)

gen err_nomd=1-nomd
gen err_bmi_nomd= bmi*err_nomd


logistic hyper  bmi err_nomd err_bmi_nomd
logit hyper  bmi err_nomd err_bmi_nomd
logit hyper  bmi nomd bmi_nomd

